

Microsoft Excel – Intersection Formulas



Intersection formulas

Intersections are like a breath of fresh air after dealing with the subtleties and complexities of functions like HLOOKUP and VLOOKUP. They are the easiest table look up formulas of all and ideal for many situations. Here is a two-variable lookup:

	Jan	Feb	Mar	Apr	May
Item1	1,496,053	1,401,484	288,202	244,217	275,347
Item2	562,075	451,313	166,374	780,827	820,455
Item3	1,486	1,409,602	522,517	457,151	498,634
Item4	1,399,143	1,303,055	1,123,933	1,459,648	16,464
Item5	8,250	335,489	1,154,902	1,106,023	1,064,783


In the above table a specific item can be returned with the following formula:

=Item2 Mar

The intersection operator is the Spacebar character. Intersection calculations do not use worksheet functions so you do not have to worry about argument orders; the previous calculation could have been entered the other way round as follows:

=Mar Item2


It is an intersection; where two references overlap. You can get problems using this simple method as the Excel program is having to guess at what we mean when we enter "Mar" or "Item2" and it does so by searching the text entered into the worksheet cells. You will have more control over your calculations if you are more deliberate in making the references using either cell references or Name references as just using the text labels can cause anomalies, especially when the same label appears more than once in different cells in the worksheet.



To name areas on the worksheet that correspond to the headings and descriptions that you have already entered, select the table of cells that you are working with and then choose *Insert, Name, Create*.

You will see that Names have now been set up in the workbook. These Names correspond to the cells which are below or to the right of the headings and descriptions around the table. We can now use these name references in our formulas.

Even if you do not enter the exact named reference in your formula you will find that the Excel formula evaluation program is quite forgiving with Name references and will usually correct any errors. Should you want to use your Name references in a formula without having to type-in the reference, then choose *Insert, Name, Paste* or press F3 and



Microsoft Excel – Intersection Formulas

select the reference from the list to insert it into your formula.

You can always see a list of the cell-related Names available in the workbook by looking in the *Name Box*. This is the drop-down list situated at the left-hand end of the Excel formula bar.

Implicit Intersections

The previous calculations were *explicit* intersections, where we gave both sides of the reference. An *implicit* intersection is where you enter only one side of the reference and the other side is implicit to the position of the formula in the worksheet.

In the example below, we can extract any monthly figure by entering just the item reference in the relevant column. Item3 entered into the same column as the Jan figures returns the Jan figure for Item3.

	Jan	Feb	Mar	Apr	May
Item1	1,496,053	1,401,484	288,202	244,217	275,347
Item2	562,075	451,313	166,374	780,827	820,455
Item3	1,486	1,409,602	522,517	457,151	498,634
Item4	1,399,143	1,303,055	1,123,933	1,459,648	16,464
Item5	8,250	335,489	1,154,902	1,106,023	1,064,783

Formulas to return the Item3 data for each month:

=Item3 =Item3 =Item3 =Item3 =Item3

Using an implicit intersection is an easy way of extracting a set of numbers from a table as all you have to do is enter the first formula and then copy to do the rest.

Indirect Intersections

In the example below, the column reference, "Mar" is entered into cell reference A8 and the row reference item, "Item4" is entered into cell A9. It has been arranged like this so that intersection formulas can be created that can be easily changed by typing a different row or column label into A8 or A9. But when we enter our intersection formula it gives us the error value, #NULL!

	A	B	C	D	E
1		Jan	Feb	Mar	Apr
2	Item1	1,496,053	1,401,484	288,202	244,217
3	Item2	562,075	451,313	166,374	780,827
4	Item3	1,486	1,409,602	522,517	457,151
5	Item4	1,399,143	1,303,055	1,123,933	1,459,648
6	Item5	8,250	335,489	1,154,902	1,106,023
7					
8	Mar	=A8 A9			
9	Item4				
10					

The formula has failed because there is no intersection between cell A8 and A9; they are both entirely separate entities and do not overlap.

Of course, that is not what we meant. We meant the references stored at cell references A8 and A9; not

Microsoft Excel – Intersection Formulas

the cells themselves but the references entered into the cells. This is an *indirect reference*. You can cure this problem easily by using the INDIRECT worksheet function. Both sides or either side in an intersection formula may be direct or indirect references.

Corrected formula:

=INDIRECT(A8) INDIRECT(A9)

To make the month reference update automatically based on the calendar, the formulas could be entered this like:

=Item2 INDIRECT(ThisMonth)

Where "ThisMonth" is a formula in a cell or a formula in a Name and the formula delivers the value in an MMM format like this:

=TEXT(NOW(),"MMM")

Or it is returned from a list, like this:

=CHOOSE(MONTH(NOW()),"Jan","Feb","Mar","Apr","May"....etc.)

Intersection references need to be more precise when they are indirect. In a direct intersection you could type in a Name like "_2005" as "2005" and it would still be accepted but the indirect reference will require the precise Name reference, otherwise it fails. The following example demonstrates a method of resolving the "underscore character in the name" issue.

	A	B	C	D	E	F	G
1	Country	Population	GDP	GDP per cap			
2	United States	266.7	7,600	28,440		Stat	GDP
3	Japan	125.7	5,100	40,500		Country	United States
4	Germany	81.3	2,500	31,170			
5	France	58.4	1,600	27,000			
6	United Kingdom	58.7	1,200	20,490		7,600	
7	Italy	56.9	1,200	20,670			
8	Brazil	164.2	883	5,230			
9	China	1,230.0	637	520			

A two-variable lookup table: cells G2 and G3 contain *Data, Validation* lists where the source ranges for the lists are the cells in row 1 for the "Stat" cell and the cells in column A for the "Country" cell. The cell formula in F6 which retrieves the data from the table is:

=INDIRECT(SUBSTITUTE(G2," ","_")) INDIRECT(SUBSTITUTE(G3," ","_"))

When we made the names for the table using the text in the headings and descriptions some of the text entries contained space characters which Excel substituted in the Name definition with an underscore character (space characters are not allowed in Names). So, when the text from the cell is used for the intersection calculation the space has to be substituted with an underscore using the SUBSTITUTE function to reference the Name correctly. The value in the Country cell is "United States" but the value required for the formula to work correctly is "United_States".

Microsoft Excel – Intersection Formulas



NB* Drop down lists in cells are very useful when you want to show a list of possible valid entries for the cell. They are very easy to do, see Data Validation is in level 3.